Dynamic Asset Allocation Strategies

Lorenzo Grassi

18/07/2025

Introduction: The Core Decision

The major decision of an investor regarding his/her portfolio is to choose the allocation between different asset classes, especially between:

- equity investments
- interest-bearing investments

Introduction: The Core Decision

Strategic asset allocation determines:

- the ultimate expected **rate of return**
- the **risk** of an investor's portfolio

Introduction: The Core Decision

Asset allocation is usually a long-term decision and its analysis should include

- wealth
- current and future cash flows
- financial goals
- inflation
- liquidity

Introduction: The Core Decision

An investor needs to determine the right balance between asset classes. This balance depends on how much risk the investor is willing to take. This amount of risk can change over time due to changes in:

- wealth
- needs
- goals
- market conditions

Impact of Time Horizon on Asset Allocation

Another important variable to consider is the investment **time horizon**. The relationship between time horizon and asset allocation has been, and still is, a topic of active debate in both theory and practice.

Time Diversification

Practitioners recommend:

- A larger allocation to stocks as the investment horizon increases (e.g., after 7 years, 95% chance of positive stock returns)
- Investing more heavily in stocks when young, due to higher human capital ("100 minus age" rule)

Samuelson–Merton Insight

The optimal proportion of wealth invested in risky assets is:

- **Constant over time**
- **Independent of current wealth**

Implication: Investment and consumption decisions are separable. Advice: maintain a fixed allocation throughout life, regardless of age or wealth.

Contrasts with practical advice (e.g., "100 – age" rule).

Single-Stage Mean-Variance Optimization

Formulation:

$$
\begin{aligned}
\min \quad & \frac{1}{2} x^\top \Sigma x \\
\text{s.t.} \quad & \mu^\top x \geq R && \text{(target expected return)} \\
& Ax = b && \text{(budget and equality constraints)} \\
& Cx \geq d && \text{(additional portfolio constraints)}
\end{aligned}
$$

Notation:

- $x \in \mathbb{R}^n$: portfolio weight vector
- $\mu \in \mathbb{R}^n$: expected return vector
- $\Sigma \in \mathbb{R}^{n \times n}$: covariance matrix of returns
- $R \in \mathbb{R}$: minimum expected return

Expected Returns and Variances

Assets considered: `stock1`, `stock2`, `bond`, `mm` (money market)

Asset	Expected Return μ	STD σ
stock1	0.09	0.15
stock2	0.07	0.11
bond	0.04	0.05
mm	0.02	0.01

Correlation and Covariance Matrices

Correlation matrix ρ:

	stock1	**stock2**	**bond**	**mm**
stock1	1.00	0.40	0.20	0.10
stock2	0.40	1.00	0.10	0.10
bond	0.20	0.10	1.00	0.30
mm	0.10	0.10	0.30	1.00

Covariance matrix $\Sigma = D\rho D$:

	stock1	**stock2**	**bond**	**mm**
stock1	0.0225	0.0066	0.0015	0.00015
stock2	0.0066	0.0121	0.00055	0.00011
bond	0.0015	0.00055	0.0025	0.00015
mm	0.00015	0.00011	0.00015	0.0001

AMPL model .dat

```
data;

set INSTR := stock1 stock2 bonds mm;


param rateofreturn :=
         stock1  0.09
         stock2  0.07
         bonds   0.04
         mm      0.02;

param covariancematrix : stock1 stock2 bonds mm :=
         stock1  0.0225  0.0066   0.0015   0.00015
         stock2  0.0066  0.0121   0.00055  0.00011
         bonds   0.0015  0.00055  0.0025   0.00015
         mm      0.00015 0.00011  0.00015  0.0001
```

AMPL model .mod

```
set INSTR;


param covariancematrix{INSTR,INSTR};
param targetR;
param rateofreturn{INSTR};

var x{INSTR} >=0;

minimize variance:
sum{i in INSTR,j in INSTR} x[i] * covariancematrix[i,j]*x[j];

s.t. cons1: sum{i in INSTR} rateofreturn[i]*x[i] >= targetR;
s.t. cons2:sum{i in INSTR} x[i]==1;
```

AMPL model .run

```
reset;

model single_stage.mod;
data single_stage.dat;

# Write CSV header
print "targetR,variance,stock1,stock2,bonds,mm" > "results_single_stage.csv";

# Set solver
option solver cplex;

# Loop through targetR values from 0.030 to 0.090 (step = 0.001)
for {t in 30..90 by 1} {
    let targetR := t / 1000;
    solve;
    printf "%.6f,%.6f,%.6f,%.6f,%.6f,%.6f\n",
        targetR,
        variance,
        x["stock1"],
        x["stock2"],
        x["bonds"],
        x["mm"] >> "results_single_stage.csv";
}
```

Efficient Portfolios

targetR	Variance	Stock1	Stock2	Bonds	MM
0.03	0.00036	0.0456	0.0735	0.1566	0.7243
0.04	0.00110	0.0902	0.1432	0.3263	0.4403
0.05	0.00231	0.1348	0.2129	0.4960	0.1563
0.06	0.00410	0.2236	0.2940	0.4824	0.0000
0.07	0.00726	0.3667	0.3889	0.2444	0.0000
0.08	0.01195	0.5097	0.4838	0.0065	0.0000
0.09	0.02250	1.0000	0.0000	0.0000	0.0000

Portfolio Composition with respect to targetR



Portfolio Composition with respect to Variance



Efficient Frontier



Scenario-Based Representation of the Stochastic Program



Tree Structure: Binary Scenario Tree

- **Binary Tree** with branching factor $b = 2$ and time horizon $T = 10$
- **Number of nodes per stage**:

$$\text{Stage } t: \quad 2^t \text{ nodes} \quad (t = 0, \ldots, T)$$

- **Node probabilities**:

$$\text{Each node at stage } t \text{ has probability } \frac{1}{2^t}$$

  - Probabilities are equal at each stage
  - Total probability across all nodes $= 1$

Objective: Wealth Evolution Over Time

- **Goal**: Simulate the evolution of portfolio wealth over a $T = 10$ period horizon
- **Initial wealth**: €100,000 at the root node (stage 0)
- **At each stage**:
  - Asset returns are applied according to the simulated AR(1) process
  - Portfolio value evolves based on the realized scenario
- **Purpose**:
  - Understand how uncertainty and correlation among asset returns affect portfolio growth
  - Support dynamic portfolio optimization across multiple periods and scenarios

Utility Functions



CARA Utility Function (Constant Absolute Risk Aversion)

Utility Function:

$$u(W) = -e^{-\gamma W}$$

Key Properties:

- $\gamma > 0$: constant coefficient of absolute risk aversion
- Absolute Risk Aversion (ARA) is constant:

$$ARA(W) = -\frac{u''(W)}{u'(W)} = \gamma$$

- Relative Risk Aversion (RRA) increases linearly with wealth:

$$RRA(W) = W \cdot \gamma$$

- Suitable when investor's risk tolerance does not depend on wealth

CARA AMPL cara.mod

```
set A;
set NODES;

param T;
param stage{NODES};
param parent{NODES} symbolic;
param prob{NODES};

param mu{A, NODES};

param gamma > 0;            # gamma is constant
param w0 >= 0;              # initial wealth

param sigma2 {A, A};

var x{A, NODES} >= 0;
var w{NODES} >= 0;

#CARA utility function
maximize Expected_Utility:
    sum {n in NODES} prob[n] * ( - exp(-gamma * w[n]) );

subject to Initial_Wealth:
    w["n0"] = w0;

subject to Initial_Budget:
    sum {a in A} x[a,"n0"] = w0;

subject to Budget {n in NODES: n != "n0"}:
    sum {a in A} x[a,n] = w[n];

subject to Wealth_Update {n in NODES: n != "n0"}:
    w[n] = sum {a in A} x[a,parent[n]] * (1 + mu[a,n]);

subject to MaxAlloc {a in A, n in NODES}:
    x[a,n] <= 0.8 * w[n];
```

Multivariate AR(1) Return Simulation on a Tree

- **Goal**: Generate return scenarios for 4 assets over a horizon $T = 10$ using a binary tree structure ($b = 2$).
- **Inputs**:
  - Mean returns μ and standard deviations σ per asset
  - Correlation matrix $\Rightarrow$ covariance matrix Σ
- **Multivariate AR(1) Model**:

$$r_{t+1} = \mu + \Phi(r_t - \mu) + \varepsilon_t, \quad \varepsilon_t \sim \mathcal{N}(0, \Sigma), \quad \Phi = 0.3 \cdot I$$

Generating AMPL .dat - AR(1) process

```python
while queue:
    parent_key, parent_name, stage, prob = queue.popleft()
    if stage >= T:
        continue
    for i in range(b):
        # new node
        child_key = parent_key + (i,)
        child_name = f"n{counter}"

        # schock
        eps = np.random.multivariate_normal(np.zeros(len(
            assets)), Sigma)

        # AR(1) simulation
        r_parent = tree[parent_name]
        r_child = mu_vec + phi @ (r_parent - mu_vec) + eps

        # saving results
        tree[child_name] = r_child
        mu[(a, child_name)] = r_child[i]
```

A Glimpse to Simulated Returns: Stock 1



Simulated Returns per Asset and Stage



Simulated Returns Density



Running the model: cara.run

```
reset;

model cara.mod;
data cara.dat;

set G := {1, 1.5, 2, 3, 4, 5, 6};

# right the header
printf "Nodo,Stage,Probability,Gamma,Wealth,stock1,stock2,bond,money\n" > "cara_gamma_all.csv";

for {g in G} {
  let gamma := g;

  option solver ipopt;
  solve;

  # append the data to the csv file
  for {n in NODES} {
    printf "%s,%d,%.4f,%.6f,%.2f", n, stage[n], prob[n], gamma, w[n] >> "cara_gamma_all.csv";
    for {a in A} {
      printf ",%.6f", x[a,n] >> "cara_gamma_all.csv";
    }
    printf "\n" >> "cara_gamma_all.csv";
  }
}
```

CARA model for different values of gamma



CARA model for different values of gamma



Summary Statistics of Final Wealth by Gamma

Gamma	Min	Q1	Median	Mean	Q3	Max	SD
1	64492	101815	112813	113126	123747	162125	15937
1.5	112594	168226	185681	185633	201622	279411	25273
2	**115942**	**172631**	**190947**	**190941**	**207900**	**290606**	**26538**
3	116922	174678	193718	193735	211201	297846	27590
4	117263	175436	194581	194629	212539	300065	27912
5	117553	175994	195452	195436	213351	302107	28215
6	117791	176476	196220	196132	214224	303892	28485

CARA model (gamma=2): Results



CARA model (gamma=2): Results



Statistic	Value
Minimum	115942
1st Quartile	172631
Median	190947
Mean	190941
3rd Quartile	207900
Maximum	290606
SD	26538

CARA model (gamma=2): Results



CARA model (gamma=2): Results



CRRA Utility Function (Constant Relative Risk Aversion)

Utility Function:

$$u(W) = \begin{cases} \frac{W^{1-\gamma}-1}{1-\gamma}, & \gamma \neq 1 \\ \log(W), & \gamma = 1 \end{cases}$$

Key Properties:

- $\gamma > 0$: constant coefficient of relative risk aversion
- Relative Risk Aversion (RRA) is constant:

$$RRA(W) = -\frac{W \cdot u''(W)}{u'(W)} = \gamma$$

- Absolute Risk Aversion (ARA) decreases with wealth:

$$ARA(W) = \frac{\gamma}{W}$$

- Reflects decreasing risk aversion as wealth increases

AMPL crra.mod

```
set A;
set NODES;

param T;
param stage{NODES};
param parent{NODES} symbolic;
param prob{NODES};

param mu{A, NODES};
param sigma2{A, A};

param gamma > 0;
param w0 >= 0;

var x{A, NODES} >= 0;
var w{NODES} >= 0;

# --- CRRA utility function ---
maximize Expected_Utility:
    sum {n in NODES} prob[n] * (
        if gamma = 1 then
            log(w[n])
        else
            (w[n]^(1 - gamma) - 1) / (1 - gamma)
    );

subject to Positive_Wealth {n in NODES}:
    w[n] >= 1e-6;

subject to Initial_Wealth:
    w["n0"] = w0;

subject to Initial_Budget:
    sum {a in A} x[a,"n0"] = w0;

subject to Budget {n in NODES: n != "n0"}:
    sum {a in A} x[a,n] = w[n];

subject to Wealth_Update {n in NODES: n != "n0"}:
    w[n] = sum {a in A} x[a,parent[n]] * (1 + mu[a,n]);

subject to MaxAlloc {a in A, n in NODES}:
    x[a,n] <= 0.8 * w[n];
```

CRRA model for different values of gamma



CRRA model for different values of gamma



Summary Statistics of Final Wealth by Gamma

Gamma	Min	Q1	Median	Mean	Q3	Max	SD
1	64492	101815	112813	113126	123747	162125	15937
1.5	112594	168226	185681	185633	201622	279411	25273
2	**115942**	**172631**	**190947**	**190941**	**207900**	**290606**	**26538**
3	116922	174678	193718	193735	211201	297846	27590
4	117263	175436	194581	194629	212539	300065	27912
5	117553	175994	195452	195436	213351	302107	28215
6	117791	176476	196220	196132	214224	303892	28485

CRRA function: 3 cases



Summary Statistics by Utility Type

Statistic	CARA	CRRA	Inc. gamma	Dec. gamma
Min	115942	126638	94340	134616
1st Quartile	172631	201169	139577	216533
Median	190947	236769	155533	259850
Mean	**190941**	**242085**	**157141**	**268688**
3rd Quartile	207900	274218	173098	305996
Max	290606	458403	232763	572628
SD	**26538**	**56962**	**23371**	**72178**

Results Comparison by Utility Type



Results Comparison by Utility Type



Results Comparison by Utility Type